Problem:
Disjoint Set

Write a program which manipulates a disjoint set S = {S1, S2, . . . , Sk}.

First of all, the program should read an integer n, then make a disjoint set where each element consists of 0, 1, ... n−1 respectively.

Next, the program should read an integer q and manipulate the set for q queries. There are two kinds of queries for different operations:

unite(x, y): unites sets that contain x and y, say Sx and Sy, into a new set.

same(x, y): determine whether x and y are in the same set.

Input

n q
com1 x1 y1
com2 x2 y2
...
comq xq yq

In the first line, n and q are given. Then, q queries are given where com represents the type of queries. '0' denotes unite and '1' denotes same operation.

Output

For each same operation, print 1 if x and y are in the same set, otherwise 0, in a line.

Constraints

1 ≤ n ≤ 10000

1 ≤ q ≤ 100000

x ≠ y

Sample Input

5 12
0 1 4
0 2 3
1 1 2
1 3 4
1 1 4
1 3 2
0 1 3
1 2 4
1 3 0
0 0 4
1 0 2
1 3 0

Sample Output

0
0
1
1
1
0
1
1

Solution in OCaml (judge verdict: Accepted)

open Printf
open Scanf

module Uftree =
  struct
    type t = int array
    let create n : t = Array.make n (-1)
    let rec find (uft : t) x =
      let p = uft.(x) in
      if p < 0 then x
      else
        let rt = find uft p in
        uft.(x) <- rt; rt
    let unite (uft : t) x y =
      let rx = find uft x in
      let ry = find uft y in
      if rx = ry then ()
      else
        if uft.(rx) < uft.(ry) then
          begin
            uft.(rx) <- uft.(rx) + uft.(ry);
            uft.(ry) <- rx
          end
        else
          begin
            uft.(ry) <- uft.(rx) + uft.(ry);
            uft.(rx) <- ry
          end
    let same (uft : t) x y =
      let rx = find uft x in
      let ry = find uft y in
      rx = ry
    let size (uft : t) x =
      let rx = find uft x in -uft.(rx)
  end

let tuple x y = (x, y)
                  
let () =
  let n, q = scanf "%d %d " tuple in
  let uft = Uftree.create n in
  let rec loop x =
    if x = 0 then ()
    else
      let com, a, b = scanf "%d %d %d " (fun i j k -> (i, j, k)) in
      if com = 0 then
        Uftree.unite uft a b
      else
        (if Uftree.same uft a b then 1 else 0) |> printf "%d\n";
      loop (x-1)
  in loop q